

Pràctiques Integrades 1er de Matemàtiques

Presentació de Maple

Curs 2003–04


Índex

I	Presentació de Maple	3
1	Càlculs Numèrics	3
1.1	Realitzant càlculs aritmètics exactes amb Maple	3
1.2	Aproximacions numèriques utilitzant la comanda <code>evalf()</code>	6
1.3	“Netejant” Variables	9
2	Càlculs Algebraics	11
2.1	La comanda <code>subs()</code>	11
2.2	La comanda <code>expand()</code>	13
2.3	La comanda <code>factor()</code>	14
2.4	La comanda <code>simplify()</code>	16
2.4.1	La comanda <code>collect()</code>	17
3	Gràfics	19
3.1	Representar una expressió: la comanda <code>plot()</code>	19
3.2	Representar diferents expressions	20
3.3	Representar punts	21
3.4	Combinar gràfics d'expressions i punts: la comanda <code>display()</code>	22
4	Resoldre equacions	25
4.1	Introduir i manipular equacions: les comandes <code>lhs()</code> i <code>rhs()</code>	25
4.2	Obtenir solucions exactes: la comanda <code>solve()</code>	26
4.3	Obtenir solucions aproximades: la comanda <code>fsolve()</code>	28
4.4	Resoldre equacions formals	31
4.5	Resoldre sistemes d'equacions lineals utilitzant la comanda <code>solve()</code>	31
5	Funcions: definir, avaluar i fer-ne gràfics	35
5.1	Definir i esborrar una funció en Maple	35
5.2	Avaluar una funció	36
5.3	Resoldre equacions en les que intervenen funcions	37

5.4	Gràfics de funcions	37
5.5	Com convertir expressions en funcions	39

Part I


Presentació de Maple

Maple és un sistema de computació algebraica molt potent. En aquesta primera secció veurem com realitzar algunes de les funcions aritmètiques bàsiques. Com veureu, les comandes s'introdueixen després del símbol $>$ que apareix per pantalla i s'acaben amb punt i coma. La línia s'executa quan premeu  i el resultat apareix al centre de la línia següent en color blau.

1 Càlculs Numèrics

En aquesta secció utilitzareu Maple per a fer alguns càlculs numèrics estàndard. L'habilitat de Maple per a donar resultats exactes a més de les aproximacions numèriques us donarà més opcions a l'hora de resoldre problemes.

1.1 Realitzant càlculs aritmètics exactes amb Maple

Utilitzar Maple per a realitzar càlculs numèrics és molt fàcil. Només s'ha d'introduir l'expressió numèrica i acabar la línia amb un **punt i coma** (;). Prement  (retorn) executareu la línia i el resultat apareixerà en color blau en el centre de la pantalla.

Exemple 1.1

Introduïu els següents càlculs simples i premeu  al final de cada línia.

$> 2+4;$

$> 12*34567890;$

$> 34+(-27);$

Cada línia vermella és “viva” i es pot modificar en qualsevol moment.

Canvieu el “4” de la línia anterior per un “8” i premeu .

Noteu com el resultat de color blau s'actualitza automàticament per a mostrar el nou resultat.

Exemple 1.2

Pel nostre exemple següent calculem 134^{39} .

```
> 134^39;
```

A diferència de la vostra calculadora, Maple us donarà la resposta exacte d'aquest problema, les 83 xifres que té.

Exemple 1.3

Maple pot fer càlculs amb fraccions sense convertir-les en expressions decimals:

```
> 3/5 + 5/9 + 7/12;
```

Exemple 1.4

Per a introduir l'arrel quadrada d'un nombre utilitzeu `sqrt()` :

```
> sqrt(24);
```

Noteu que Maple ha simplificat $\sqrt{24}$ però que ha deixat el resultat en forma exacta. En el proper apartat aprendreu com obtenir una aproximació decimal per a aquest nombre.

Exemple 1.5

Maple té introduïdes totes les constants matemàtiques importants. Per a introduir π escriviu `Pi`.

Fixeu-vos que es necessita un asterisc `*` per a indicar la multiplicació.

```
> 4*(3+Pi);
```

Una altra vegada Maple fa el càlcul però deixa el resultat en una forma exacta.

Exemple 1.6

A diferència de la vostra calculadora, Maple dona la solució exacta quan s'aplica a les funcions trigonomètriques.

```
> sin(5*Pi/3);  
> sec(Pi/4);
```

Per a obtenir l'invers del sinus d'un nombre utilitzeu la funció `arcsin()`:

```
> arcsin(-1);
```

Si demaneu a Maple que calculi un valor *indeterminat* respondrà amb un missatge d'error:

```
> tan(Pi/2);
```

Exemple 1.7

Per a introduir la funció exponencial e^x en Maple escriviu: `exp(x)` .

```
> exp(x);
```

I per a obtenir el nombre e escriviu: `exp(1)` .

```
> exp(1);
```

Exemple 1.8

Per a introduir la funció valor absolut $|x|$ en Maple escriviu: `abs(x)`.

Noteu que Maple dona la solució correcta i exacta per a la tercera línia ja que: $e - \pi < 0$


```
> abs(x);  
> abs(-3);  
> abs(exp(1)-Pi);
```

Exemple 1.9

Maple té moltes comandes per a realitzar càlculs específics amb nombres. Les podreu aprendre així que es vagin necessitant. Aquí presentem un últim exemple per ara: si teniu un nombre enter i voleu obtenir la seva descomposició en factors primers podeu utilitzar la comanda de Maple `ifactor()`. Podeu practicar tot el que vulgueu canviant el nombre que hi ha aquí sota.

```
> ifactor(31722722304);
```

Exemple 1.10

Pot haver hi molts moments en els que vulgueu introduir més d'una comanda en la mateixa línia. Això es pot fer en Maple, només cal assegurar-se d'acabar **cada una** de les comandes amb un punt i coma (;). També ajudarà posar espais en blanc entre les comandes. Quan premeu  totes les expressions s'executaran i els resultats apareixeran, en ordre, en un únic camp de resultats.

```
> sin(Pi/3);   cos(Pi/3);   tan(Pi/3);
```

Exemple 1.11

Per a calcular i mostrar una successió de nombres utilitzeu la comanda `seq()`. Aquí mostrem els 100 primers nombres naturals, els seus quadrats i els cent primers nombres imparells.

```
> seq(n,n=1..100);
> seq(k^2,k=1..100);
> seq(2*k+1,k=1..100);
```

1.2 Aproximacions numèriques utilitzant la comanda `evalf()`

Recordeu que en l'apartat anterior hem demanat a Maple que fes la suma de tres fraccions i que ha donat com a resultat una altra fracció. Aquest tipus d'aritmètica exacta és molt útil però hi ha vegades que preferirem una resposta en forma d'expressió decimal. La comanda de Maple `evalf()` realitza aquesta tasca per nosaltres.

Exemple 1.12

Compareu els resultats de les dues línies següents.

```
> 3/5+5/9+7/12;
> evalf(3/5+5/9+7/12);
```

Exemple 1.13

Donar un nom al resultat d'un determinat càlcul fa que sigui més fàcil utilitzar-lo en un càlcul posterior. Per assignar un nom utilitzem uns **dos punts** seguit del **signe igual** (és a dir `nom := resultat`). En la línia que ve a continuació hem assignat a la lletra **k** el resultat del càlcul anterior. Després apliquem `evalf()` a **k**.

```
> k:=3/5+5/9+7/12;  
> evalf(k);
```

Nota important de Maple: Maple distingeix entre majúscules i minúscules. Per tant Maple considera que k i K són variables diferents.

```
> k;  
> K;
```

Per cert també es poden utilitzar paraules com a nom de variables.

```
> josep:=2^5;  
> sqrt(josep);
```

Exemple 1.14

Si volem més o menys dígit de precisió en comptes dels 10 que s'utilitzen per defecte podem afegir un argument extra a la comanda `evalf()` tal i com es mostra a continuació.

```
> w:=4*(3+Pi);  
> evalf(w);  
> evalf(w,4);  
> evalf(w,45);
```

Exemple 1.15

Si introduïu nombres amb un punt decimal Maple donarà de forma automàtica un resultat decimal. Compareu els dos resultats que apareixen a continuació.

```
> sqrt(34);  
> sqrt(34.0);
```

Aquí hi ha un altre exemple:

```
> 4-1/3;  
> 4.0-1/3;
```

Exemple 1.16

Podem aplicar la comanda `evalf()` a una successió de nombres: aquí sota generem primer les arrels quadrades exactes dels 10 primers nombres naturals, després apliquem `evalf()` per a obtenir les aproximacions decimals. Recordeu la comanda per crear successions de números que hem après a l'Exemple 1.11.

```
> result:=seq(sqrt(k),k=1..10);  
> evalf(result);
```

Drecera de Maple: referir-se de manera ràpida a l'últim resultat

Quan utilitzeu Maple hi ha moltes vegades que voldreu encadenar una sèrie de càlculs. En comptes de donar un nom a cada un dels resultats que aneu obtenint, podeu utilitzar el signe tant per cent (`%`) per referir-vos a **l'última expressió que Maple ha calculat**. Aquí hi ha alguns exemples de com funciona.

```
> 3/5+5/9+7/12;  
> evalf(%);  
> Pi;  
> evalf(%);  
> %+5;
```

Per a més informació de com fer servir aquest símbol mireu l'Exercici 1.4 que ve tot seguit.

Exercici 1.1

Utilitzeu Maple per a calcular el nombre 37^{43} .

Exercici 1.2

Calculeu $\sqrt{34}$ amb 18 xifres.

Exercici 1.3

Obteniu una aproximació numèrica per a l'expressió : $\frac{3 + \pi}{7 - \sqrt{13}}$

Exercici 1.4

El símbol tant per cent (%) és una drecera molt útil però pot donar en alguns casos resultats inesperats. Aquí hi ha un exemple.

Primer executeu cada una de les tres línies següents. Heu de poder dir quin serà el resultat per endavant.

```
> 4+Pi;  
> evalf(%);  
> %+10;
```

Ara torneu enrera i torneu a executar l'última línia (i.e. > %+10;). Fixeu-vos que el resultat canvia de 17.14159265 a 27.14159265

Podeu explicar el perquè?

1.3 “Netejant” Variables

Des del moment en que definiu una variable, Maple recordarà quin és el seu valor durant tota la sessió de treball. Si voleu introduir un nou valor a la variable, només heu de fer la nova assignació.

Per exemple cada una de les assignacions que hi ha aquí baix redefineix el valor de la variable **h**. (Nota: per a verificar el valor actual d'una variable només cal escriure el seu nom seguit d'un punt i coma).

```
> h;  
> h:=56;  
> h;  
> h:=sqrt(Pi);  
> h;
```

A vegades voldreu “netejar” una variable de la memòria per a poder utilitzar-la en una situació nova. Aquí hi ha un exemple. Primer introduïm en **x** el valor 65.

```
> x:=65;
```

Ara suposem que comencem un problema nou i que volem introduir l'expressió algebraica genèrica (depenent d'una variable indeterminada x) $x^2 - 4x + 7$ i assignar-li el nom **w**. Si només fem això, Maple substitueix automàticament el valor anterior de **x**.

```
> w:=x^2-4*x+7;
```

Per a fer que x sigui una variable genèrica un altre cop hem de “netejar” (és a dir esborrar la memòria de Maple) el nostre antic valor per a x . Això s’aconsegueix introduint

```
> x := 'x';
```

Fixeu-vos que aquí utilitzem una cometa (apòstrof).

Executeu les dues línies següents per a veure com funciona.

```
> x:='x';
```

```
> w:=x^2-4*x+7;
```

Netejant totes les variables de cop: la comanda `restart`.

La comanda `restart` neteja la memòria de Maple de totes les definicions que hagueu fet. És com si es comencés una nova sessió de Maple. Si heu de començar un problema totalment nou podeu utilitzar la comanda `restart` per assegurar que no queden definicions del vostre treball anterior. Abans d’executar la segona línia d’aquí sota, endevineu ràpidament el resultat.

```
> p:=4;
```

```
> p; x; h;
```

Probablement recordareu que p era 4 i que x s’havia reassignat a la variable genèrica x , però poder no haureu recordat que h s’havia assignat al valor $\sqrt{\pi}$. És per això que és una bona idea utilitzar `restart` per a eliminar totes les definicions d’un sol cop. (Si aneu seguint aquestes pràctiques veureu que començarem moltes seccions noves amb una comanda `restart`).

```
> restart;
```

```
> p; x; h;
```

2 Càlculs Algebraics

Maple és un “C.A.S” , això vol dir un “**Computer Algebra System**” (que en català s’anomena normalment un *Manipulador Algebraic*). Això significa que Maple coneix totes les regles algebraiques que vosaltres sabeu. Al mateix temps que aneu progressant en el Càlcul, l’Àlgebra lineal i les altres especialitats matemàtiques veureu que Maple també conté les operacions essencials d’aquests temes introduïdes en el seu gran conjunt de comandes.

En aquesta secció aprendreu com introduir una expressió algebraica i a donar valors a les seves variables. Després aprendreu les comandes que us permetran expandir, factoritzar i simplificar expressions.

És recomanable iniciar les sessions executant la comanda `restart` que ja hem vist a la secció anterior, recordeu quin és el seu efecte?

```
> restart;
```

2.1 La comanda `subs()`

La primera comanda que analitzarem ens permetrà substituir valors en les variables o paràmetres d’una expressió algebraica que haguem definit.

Exemple 2.1

Com a primer exemple comencem amb l’expressió $3x^2+8$ i assignem-li com a nom `W`.

```
> W:=3*x^2+8;
```

```
> W;
```

Ara suposem que volem substituir la variable `x` de l’expressió $3x^2+8$ pel valor 4. La forma més ràpida de fer-ho és utilitzar la comanda de Maple `subs()`. Tot seguit veiem com fer-ho.

```
> subs(x=4,3*x^2+8);
```

De forma alternativa podem aplicar la comanda `subs()` a `W`.

```
> subs(x=4,W);
```

Exemple 2.2

La comanda `subs()` també funciona bé amb valors simbòlics. Per exemple, per a substituir `x` per $5 + 2u$ en l’expressió $3x^2 + 8$ executem la línia següent (en aquest cas posem l’etiqueta `M` al resultat).

```
> W:=3*x^2+8;
```

```
> M:=subs(x=5+2*u,W);
```

I ara per a fer que Maple “multipliqui” aquesta expressió utilitzem la comanda `expand()`.

```
> expand(M);
```

Exemple 2.3

La comanda `subs()` és molt versàtil. Podem utilitzar-la per a avaluar expressions en les que intervenen més d'una variable. Aquí substituïm la x per 7 i la y per 12 en l'expressió $U = \frac{2x^2}{5} + 3y$.

```
> U:=(2/5)*x^2+3*y;
> subs(x=7,y=12,U);
> evalf(%);
```

Exemple 2.4

També podem utilitzar la comanda `subs()` per a substituir un valor en una equació. Això és el tipus de coses que cal fer per a verificar si un determinat valor “satisfà” l'equació. En les línies que venen a continuació substituïrem diferents valors en l'equació $x^3 - 5x^2 + 7x - 12 = 0$. És algun d'aquests valors una solució de l'equació?

Noteu que utilitzem “:=” per assignar el nom i només “=” per a l'equació mateixa.

```
> eqn:=x^3-5*x^2+7*x-12=0;
> subs(x=3,eqn);
> subs(x=4,eqn);
> subs(x=5,eqn);
```

Exercici 2.1

Assigneu el nom k a l'expressió $x^2 + 4x - 3$. Un cop fet això dieu M a l'expressió $k^2 - 9$. Finalment feu que Maple calculi $3M + 6$.

Nota: per fer que Maple multipliqui les expressions utilitzeu la comanda `expand()`. És a dir, introduïu

```
> expand(3*M+6);
```

Apreneu més coses sobre la comanda `expand()` en l'apartat següent.

Exercici 2.2

Desenvolueu $(1 + x)^4$ utilitzant la comanda `expand()`.

Exercici 2.3

Sigui $P = ax^3 + bx^2 + cx + d$. Calculeu P si $x = 0.01$, $a = -\frac{1}{5}$, $b = \frac{2}{5}$, $c = 0$, i $d = \frac{13}{15}$

Exercici 2.4

Utilitzeu la comanda `subs()` per a verificar si algun dels nombres 1, 2 o 3 és una solució de l'equació $x^3 - 16x^2 + 51x - 36 = 0$.

2.2 La comanda `expand()`

L'ús principal de la comanda `expand()` és el de “fer les multiplicacions” en els productes d'expressions polinòmiques. També es pot utilitzar per a desenvolupar funcions trigonomètriques i altres tipus de funcions més generals.

Exemple 2.5

Utilitzeu la comanda `expand()` per efectuar les multiplicacions en $(x + 2)^2 (3x - 3)(x + 5)$.

```
> k:=(x+2)^2*(3*x-3)*(x+5);  
> expand(k);
```

Exemple 2.6

Maple aplica algunes de les identitats trigonomètriques més comuns per a desenvolupar $\sin(2x)$ i $\cos(2x)$.

```
> expand(sin(2*x));  
> expand(cos(2*x));
```

Feu alguna prova desenvolupant el sinus i el cosinus d'alguns altres múltiples enters de x . Per exemple: $\sin(3x)$, $\cos(6x)$, etc.

Exemple 2.7

Finalment feu que Maple multipliqui l'expressió $x^{\frac{1}{2}}(x^{\frac{3}{2}} + x^{-\frac{1}{2}})$

```
> h:=x^(1/2)*(x^(3/2)+x^(-1/2));
> expand(h);
```

Exercici 2.5

Desenvolpeu $(x + 1)^n$ per a $n = 2, 3$ i 4 .

2.3 La comanda factor()

La comanda `factor` agrupa un polinomi en producte de factors utilitzant nombres enters i fraccions. Aprendre el seu funcionament amb una serie d'exemples.

Exemple 2.8

Factoritzem l'expressió: $3x^2 - 10x - 8$

```
> w:=3*x^2-10*x-8;
> factor(w);
```

O podem fer el mateix en una sola línia:

```
> factor(3*x^2-10*x-8);
```

Exemple 2.9

Primer desenvolupem l'expressió $2(x - 2)(2x^2 + 5x + 2)(x + 4)$. Després, per a tornar a descomposar l'expressió, apliquem la comanda `factor()` al resultat. Podeu explicar per què el resultat final sembla diferent de l'expressió original?

```
> H:=2*(x-2)*(2*x^2+5*x+2)*(x+4);
> ans:=expand(H);
> factor(ans);
```

Exemple 2.10

Maple pot factoritzar expressions amb més d'una variable. Per exemple, factoritzem l'expressió, donada en funció de les variables x i y , $x^2y + 2xy + y$

```
> h:=x^2*y+2*x*y+y;  
> factor(h);
```

Exemple 2.11

Si Maple no pot factoritzar una expressió utilitzant nombres racionals (és a dir, enters i fraccions) donarà com a resultat el mateix que heu introduït sense cap canvi.

```
> factor(3*x^2-10*x-9);
```

Exemple 2.12

La comanda `factor()` no està limitada als polinomis. Es pot utilitzar per a factoritzar altres formes d'expressions, per exemple, les que contenen funcions trigonomètriques. Factoritzeu $(\sin x)^2 - (\cos x)^2$.

```
> factor((sin(x))^2-(cos(x))^2);
```

Exemple 2.13

Si la comanda `factor()` s'utilitza amb una expressió racional, es factoritza el numerador i el denominador i els factors comuns es cancel·len per a simplificar l'expressió:

```
> A:=(x^3-7*x^2+15*x-9)/(x^2+4*x+4);  
> factor(A);  
> B:=(x^3-7*x^2+15*x-9)/(x^2-4*x+3);  
> factor(B);
```

El següent exemple us permet veure la forma factoritzada sense les simplificacions.

Exemple 2.14

Les comandes de Maple `numer()` i `denom()` us permeten aïllar tant el numerador com el denominador d'una fracció. Aquí utilitzem aquestes comandes per examinar els factors del numerador i el denominador per separat (és a dir, abans de les simplificacions dels factors comuns).

```
> B:=(x^3-7*x^2+15*x-9)/(x^2-4*x+3);
> numer(B); denom(B);
> factor(numer(B)); factor(denom(B));
```

Exercici 2.6

Factoritzeu l'expressió $3x^4 - 2x^3 + 22x^2 - 18x - 45$.

Exercici 2.7

Factoritzeu l'expressió $x^{\frac{1}{2}} - x^{\frac{3}{2}}$ i després utilitzeu la comanda `expand()` per a comprovar el resultat.

2.4 La comanda `simplify()`

La comanda `simplify()` és la que ens permet simplificar diferents tipus d'expressions. Comencem veient el seu funcionament amb un seguit d'exemples.

Exemple 2.15

Considerem l'expressió $(\cos x)^5 + (\sin x)^4 + 2(\cos x)^2 - 2(\sin x)^2 - \cos(2x)$. Maple pot aplicar identitats per a simplificar expressions matemàtiques llargues, com ara expressions trigonomètriques.

```
> V:=cos(x)^5 + sin(x)^4 + 2*cos(x)^2 - 2*sin(x)^2 - cos(2*x);
> simplify(V);
```

Exemple 2.16

Les expressions trigonomètriques amb arguments donats com a múltiples d'algun angle queden simplificades en termes de funcions trigonomètriques d'aquest angle si és possible:

```
> simplify(sin(5*t)+sin(3*t));
```

Exemple 2.17

La comanda `simplify()` es pot utilitzar per a sumar expressions racionals. A continuació reescrivim la suma $\frac{1}{x+1} + \frac{x}{x-1}$ com una única fracció.

```
> M:=(1/(x+1))+(x/(x-1));
```

```
> simplify(M);
```

Exercici 2.8

Simplifiqueu l'expressió $\frac{7}{x+2} + \frac{3x}{(x+2)^2}$

Exercici 2.9

Com simplifica Maple l'expressió $\sin(3t) - \sin(7t)$? Si aquesta expressió “simplificada” és o no és útil dependrà del que es tingui planejat fer amb ella.

2.4.1 La comanda `collect()`

Hi ha ocasions en les que el que es necessita no és una simplificació com les que dona la comanda `simplify()` si no que el que interessa és *agrupar* els termes d'una expressió segons les potències (enteres o fraccionaries) d'una variable en concret. Per fer aquest tipus de manipulació es pot utilitzar la comanda `collect()` com en l'exemple següent:

Exemple 2.18

Si considereu l'expressió polinòmica $x^3 - x^2 y + x^2 - 2xy - x - y^2 x + y^3 + y^2 - y - 1$ (respecte les variables x i y) podeu agrupar respecte x amb

```
> collect(x^3-x^2*y+ x^2-2*x*y-x-y^2*x+y^3+y^2-y-1,x);
```

Com seria l'expressió agrupant respecte de la variable y ?

Si mireu en l'ajuda de Maple veureu que la comanda `collect()` admet que la *variable* respecte de la que es fa l'agrupament pugui ser una expressió gairebé arbitrària.

Exercici 2.10

Agrupeu en funció de $\ln(x)$ l'expressió $a \ln(x) - x \ln(x) - x$. Feu el mateix, respecte e^x , amb l'expressió $x^2 e^x - 2x e^x + 2e^x - \frac{x^2}{e^x} - 2\frac{x}{e^x} - \frac{2}{e^x}$.

3 Gràfics

En aquesta secció aprendreu a dibuixar el gràfic d'una funció definida per una expressió. A més, altres temes que hi podeu trobar inclouen: combinar gràfics de diferents expressions en un únic dibuix, representar punts, i combinar diferents estructures gràfiques en un únic dibuix.

```
> restart;
```

3.1 Representar una expressió: la comanda `plot()`

La comanda bàsica per representar gràficament una expressió o funció d'una variable és `plot`. Aprendrem el seu funcionament a partir dels exemples següents.

Exemple 3.1


Utilitzem la comanda `plot()` per a dibuixar el gràfic de $3x^2 - 8$ per a x entre -5 i 5 .

```
> plot(3*x^2-8,x=-5..5);
```

Observeu que Maple posa l'escala de l'eix de les y de manera automàtica, triant l'escala de les y que mostra tot el gràfic que correspon al domini que s'ha especificat.

Podem eliminar l'elecció automàtica de l'escala de les y especificant a part del rang de les x un rang per a les y . En la línia següent hem limitat el rang de les y a l'interval $[-20, 40]$.

```
> plot(3*x^2-8,x=-5..5,-20..40);
```

Si feu clic amb el botó esquerre del ratolí, el gràfic queda seleccionat i la barra d'opcions inferior es modifica. Ara quan feu clic en el gràfic, les coordenades del punt del lloc on ho feu es veuran en la finestra de l'esquerra. El botó  fa que les escales de les x i de les y siguin iguals.

Podem experimentar sobre el gràfic anterior les diferents opcions possibles.

Exemple 3.2

L'escala automàtica és una característica útil però a vegades necessitarem especificar manualment el rang de les y . Per exemple l'escalat automàtic no és apropiat per a gràfics amb asímptotes verticals.

Compareu els dos gràfics següents. Noteu que hem fixat els límits per a les y a l'interval $[-20, 20]$ en la segona comanda `plot()`.

```
> plot(x/(x-2),x=-5..5);
```

```
> plot(x/(x-2),x=-5..5,-20..20);
```

Exemple 3.3

Dibuixarem el gràfic de $y = x^3 + 1 - e^x$ en el domini $[-8, 8]$. Triarem un rang per a les y que permeti veure els quatre talls amb l'eix de les x .

Primer doneu un cop d'ull al gràfic amb l'escala automàtica per a les y .

```
> plot(x^3+1-exp(x),x=-8..8);
```

Com que els valors de les y (valors de l'expressió) a prop de 8 són negatius i molt grans en valor absolut l'escala vertical ha hagut de ser massa gran per a veure clarament els talls amb l'eix de les x .

S'obté una visió molt millor restringint els límits en el rang de les y .

```
> plot(x^3+1-exp(x),x=-8..8,-5..15);
```

Exercici 3.1

Dibuixeu $y = \sin(x)$ per a dos períodes complets.

Exercici 3.2

Dibuixeu $y = 3x^4 - 6x^2$ per al domini $[-10, 10]$ amb escala automàtica per a les y . Després d'observar el gràfic, editeu el domini i el recorregut per tal de veure els talls amb l'eix de les x clarament. Feu una estimació dels talls amb l'eix x amb el cursor.

3.2 Representar diferents expressions

Per a mostrar més d'un gràfic en el mateix dibuix feu-ne la llista dins de claudàtors [] separant-los per comes. Per exemple,

```
> plot([cos(x),x^2],x=-1..4,-4..4);
```

Noteu que cada un dels gràfics es mostra utilitzant un color diferent. Podeu especificar els colors per a cada funció afegint una opció de color al final de la comanda. Els colors s'assignaran en el mateix ordre que el de les funcions. Fixeu-vos que la llista dels colors també es fa amb uns claudàtors []. Aquí hi ha un exemple.

```
> plot([cos(x),x^2],x=-1..5,-4..4, color=[blue,black]);
```

Aquests són els colors disponibles en Maple (s'han d'escriure en anglès, no val posar **groc**):

aquamarine, black, blue, navy, coral, cyan, brown, gold, green, gray, grey, khaki, magenta, maroon, orange, pink, plum, red, sienna, tan, turquoise, violet, wheat, white i yellow.

Exercici 3.3

Feu junts els gràfics de les funcions $y = x^2 - 5x + 6$ i $y = \frac{1}{(x-2)^2}$. Experimenteu amb diferents rangs per a les y de forma que es puguin veure dibuixos complets dels dos gràfics.

3.3 Representar punts

La comanda `plot()` pot dibuixar també un o més punts.

Exemple 3.4

Dibuixem el punt (2,3). Noteu en la línia següent que utilitzem dos jocs de claudàtors.

```
> plot([ [2,3] ],style=point);
```

Podem controlar la mida dels rangs per a les x i per a les y afegint aquesta informació a la comanda com en la línia següent.

```
> plot([ [2,3] ],-7..7,-7..7,style=point);
```

Exemple 3.5

Per a dibuixar més d'un punt fem una llista dins la comanda `plot()` (observeu les comes). Recordeu que s'ha de posar un parell de claudàtors per cada punt i un parell extra envoltant la llista.

```
> plot([ [2,3],[ -2,5],[ 1,-4] ],-7..7,-7..7, style=point);
```

Exemple 3.6

Canviant l'estil a `line` es connecten els punts conservant l'ordre de la llista.

```
> plot([ [2,3],[ -2,5],[ 1,-4] ],-7..7,-7..7, style=line);
```

Què passa si canviem l'ordre dels punts de la llista?

Exemple 3.7

Es poden utilitzar extensions opcionals per a especificar el color dels punts i el símbol que es fa servir (per exemple `diamond`, `circle` i `cross`, que és el que hi ha per defecte) per a representar-los. Utilitzant l'ajuda de Maple, esbrineu les diferents possibilitats que admet l'opció `symbol`.

```
> plot([[3,2],[2,-1]],style=point,color=blue,symbol=circle);
```

Exercici 3.4

Dibuixeu els punts següents utilitzant el color vermell i el símbol `diamond`: $[1, 4]$, $[-2, -3]$, $[4, -5]$ i $[-6, 5]$. Després connecteu els punts amb línies rectes amb una comanda `plot()` a part.

3.4 Combinar gràfics d'expressions i punts: la comanda `display()`

Hi ha comandes de Maple que no estan disponibles fins que no es *carrega* un *paquet* que les activa. Un d'aquests paquets és el **plots** que estén les possibilitats gràfiques. Per accedir a les comandes d'un paquet (carregar el paquet) s'utilitza la comanda `with()`, la línia següent carrega el paquet **plots**

```
> with(plots);
```

Després d'executar aquesta comanda veureu que Maple ha mostrat, com a resultat, el nom de les comandes noves que proporciona el paquet. Si no volem que el resultat d'una comanda es mostri per pantalla però sí que s'executi, podem acabar la línia amb dos punts enlloc de punt i coma. Els dos punts al final de la línia fa que la línia es pugui executar sense mostrar cap resultat i serveix per a qualsevol comanda.

```
> with(plots):
```

De les comandes noves que apareixen quan carreguem **plots** mencionarem (de moment) únicament la comanda `display()` que permet combinar gràfics d'expressions i de punts i línies en el mateix dibuix. Per utilitzar `display()` el primer pas consisteix en nomenar individualment cada un dels components del dibuix.

Important: Ens hem d'assegurar que utilitzem uns **dos punts** al final de cada línia per a suprimir la presentació dels resultats ja que si no es fa el que s'obté com a resultat és força desagradable (mireu les tres primeres línies que hi ha a sota).

Després la comanda `display()` s'utilitza per a fer el dibuix que volem (acaba amb un punt i coma).

```
> pict1:=plot([-3*x+5,9-x^2],x=-3..5,color=[green,red]):
> pict2:=plot([[1,8],[4,-7]],style=point,color=blue,symbol=circle):
> display([pict1,pict2]);
```

De forma alternativa podem fer la llista d'aquestes tres comandes `plot()` relacionades en un sol grup d'execució. Com que sovint sortiran línies bastant llargues és útil canviar de línia sense executar la comanda prement les tecles `Shift ↑` + `←` a l'hora en el punt on volem que acabi cada una de les línies.

```
> pict1:=plot([-3*x+5,9-x^2],x=-3..5,color=[green,red]):  
  pict2:=plot([-1,8],[4,-7],style=point,color=blue,symbol=circle):  
  display([pict1,pict2]);
```

Exercici 3.5

Feu un gràfic que contingui a l'hora el gràfic de la funció $y = x^2 + x - 6$ i les seves interseccions amb l'eix de les x i el de les y (per a $x = -3$ i $x = 2$) marcades amb un cercle. (Podeu consultar en l'ajuda de Maple com la comanda `solve()`, que introduïrem en una secció propera, us pot donar aquests valors de la x).

4 Resoldre equacions

En aquesta secció aprendrem a aplicar la comanda de Maple `solve()` per a obtenir les solucions **exactes** d'equacions (quan això sigui possible). Recordeu que en nombrosos casos no és possible obtenir solucions exactes de les equacions i que s'ha de deixar la feina als *solucionadors numèrics* per a poder calcular solucions **aproximades**. Més endavant en aquesta mateixa secció utilitzarem la comanda de Maple `fsolve()` per a obtenir aproximacions decimals de solucions. També es discutirà la resolució de sistemes d'equacions lineals.

Comencem reinicialitzant les variables i carregant el paquet **plots**.

```
> restart;  
> with(plots):
```

4.1 Introduir i manipular equacions: les comandes `lhs()` i `rhs()`

En aquest apartat veurem unes comandes bàsiques que ens permeten manipular equacions i extreure'n informació.

Exemple 4.1

Es pot donar un nom a una equació igual que es fa amb qualsevol altre expressió. En la línia següent introduïm l'equació $x^3 - 5x^2 + 23 = 2x^2 + 4x - 8$ i li posem com a nom `eqn1`.

```
> eqn1:=x^3-5*x^2+23=2*x^2+4*x-8;  
> eqn1;
```

Exemple 4.2

Podem aïllar la part esquerra i la part dreta de l'equació utilitzant les comandes `lhs()` i `rhs()`.

```
> lhs(eqn1);  
> rhs(eqn1);
```

Exemple 4.3

Vegeu com podem utilitzar les comandes `lhs()` i `rhs()` per a obtenir una equació que és equivalent a l'equació original `eqn1` però que té com a part dreta un 0. Posem-li com etiqueta `eqn2`.

```
> eqn2:=lhs(eqn1)-rhs(eqn1)=0;  
> eqn2;
```

4.2 Obtenir solucions exactes: la comanda `solve()`

En aquest apartat veurem com utilitzar la comanda `solve` en diferents situacions i el seu funcionament. Per començar considerarem equacions polinòmiques. Existeixen algorismes per a calcular les solucions **exactes** per a polinomis que tenen fins a **grau 4**. La comanda de Maple `solve()` coneix aquests algorismes i els aplica. Aquesta comanda necessita dos arguments: l'equació i el nom de la variable.

Exemple 4.4

Per a obtenir les solucions exactes de l'equació polinòmica $3x^3 - 4x^2 - 43x + 84 = 0$ utilitzem la comanda `solve()`. Noteu que el segon argument de la comanda li diu a Maple que x és la incògnita respecte de la qual volem obtenir la solució.

```
> solve(3*x^3-4*x^2-43*x+84=0,x);
```

Aquí Maple ha trobat les tres solucions i les ha ensenyades. Les solucions són retornades agrupades en una estructura de dades amb les diferents solucions separades per comes.

Exemple 4.5

A vegades voldrem seleccionar una solució de la llista de solucions obtinguda i utilitzar-la posteriorment en un altre càlcul. Podem fer això assignant primer un nom (en aquest cas utilitzem la lletra `N`) al resultat obtingut de la comanda `solve()`. Aleshores podem utilitzar un índex numèric per aïllar elements de la llista. Per exemple `N[1]` és el primer número de la llista, `N[2]` és el segon i així successivament (fixeu-vos que utilitzem claudàtors).

```
> N:=solve(x^2-5*x+3=0,x);
```

```
> N[1];
```

Exemple 4.6

Quan es treballa amb la comanda `solve()` sovint és convenient començar donant un nom a l'equació. Recordeu que utilitzem `:=` per assignar el nom i només `=` per a l'equació pròpiament dita.

```
> eqn1:=7*x^3-11*x^2-27*x-9=0;
```

Seguidament resollem l'equació utilitzant la comanda `solve()` i assignem el nom `H` al resultat.

```
> H:=solve(eqn1,x);
```

Per a practicar comproveu que cada un d'aquests valors satisfà l'equació. Això es pot fer fàcilment utilitzant la comanda `subs()` que ja ha estat analitzada en una secció anterior. Repaseu-ne la sintaxi i el seu funcionament.

```
> subs(x=H[1],eqn1);
```

```
> subs(x=H[2],eqn1);
> subs(x=H[3],eqn1);
```

Exemple 4.7

A vegades les solucions “exactes” són massa complicades per a poder ser realment útils. En les dues línies següents resollem l'equació $x^3 - 34x^2 + 4 = 0$.

```
> eqn1:=x^3-34*x^2+4=0;
> H:=solve(eqn1,x);
```

Com podeu veure, llegir aquestes solucions exactes és realment un repte! Noteu que I representa $\sqrt{-1}$. Quan una solució és així de complicada és més útil mirar una solució aproximada utilitzant la comanda `evalf()` que ja coneixeu.

```
> evalf(H);
```

Una bona alternativa a la comanda `solve()` en una situació com la de l'exemple que acabem de veure és la comanda `fsolve()` que discutirem en la secció següent.

La comanda `solve()` també es pot utilitzar per a determinar solucions exactes d'equacions **no polinòmiques**. Tot seguit veurem una llista d'alguns exemples simples d'aquest tipus. En tot cas, si les equacions són complicades, per exemple si combinen exponencials, polinomis i expressions trigonomètriques, normalment no es podrà disposar de les solucions exactes. Un altre cop la comanda `fsolve()` serà una alternativa.

Exemple 4.8

Resoleu l'equació $5e^{\frac{x}{4}} = 43$.

```
> solve(5*exp(x/4)=43,x);
```

Exemple 4.9

A vegades Maple no mostra **totes** les solucions possibles.

```
> solve(sin(x)=1/2,x);
```

Com podeu utilitzar el resultat que acabeu d'obtenir amb la comanda `solve` per a poder escriure el conjunt de totes les solucions de l'equació?

Exercici 4.1

Resoleu l'equació $x^3 - 11x^2 + 7x + 147 = 0$. Per què Maple produeix només dues solucions diferents per a aquesta equació cúbica? Per què una d'elles està escrita dues vegades? (**Indicació:** Factoritzeu la part esquerra de l'equació).

El fet que $(x - 7)$ és un **factor repetit** porta com a conseqüència que l'equació cúbica tingui només dues solucions diferents -3 i 7 . Diem que l'arrel 7 té **multiplicitat 2**, això significa que hi ha dos factors de la forma $(x - 7)$ en la factorització del polinomi.

4.3 Obtenir solucions aproximades: la comanda `fsolve()`

La comanda `fsolve()` es pot utilitzar per a aproximar solucions de qualsevol tipus d'equació i funciona de manera anàloga a la comanda `solve()`. Per a equacions polinòmiques `fsolve()` produeix una **llista completa** de totes les solucions reals en un sol pas (mireu l'Exemple 4.10). Per a altres tipus d'equacions, `fsolve()` es pot utilitzar per a obtenir les solucions **d'una en una** (mireu els Exemples 4.11 i 4.12).

Exemple 4.10

La comanda de Maple `fsolve()` calcularà una aproximació numèrica per a cada una de les solucions reals d'una equació polinòmica. En aquest exemple l'utilitzarem per aproximar totes les solucions reals de l'equació: $x^4 - x^3 - 17x^2 - 6x + 2 = 0$.

```
> eqn:=x^4-x^3-17*x^2-6*x+2=0;
> fsolve(eqn,x);
```

Les quatre solucions que es mostren ens donen una llista completa de les aproximacions de les solucions de l'equació polinòmica.

Exemple 4.11

En aquest exemple trobarem **totes** les solucions reals de l'equació $x^3 + 1 - e^x = 0$ utilitzant la comanda `fsolve()`.

```
> eqn:=x^3+1-exp(x)=0;
> fsolve(eqn,x);
```

Maple ens dona una solució real però aquest cop Maple no ens ha explicat la història completa. Hi ha altres solucions? Com es poden trobar? Tot seguit en l'Exemple 4.12 es presenta un procediment sistemàtic per a determinar les solucions que resten.

Exemple 4.12

Troblem les altres solucions reals de l'equació $x^3 + 1 - e^x = 0$. El primer pas per a trobar les altres solucions és fer un dibuix del gràfic de la part esquerra de l'equació. Recordeu que els talls amb l'eix de les x de $y = x^3 + 1 - e^x$ es corresponen exactament amb les solucions de l'equació $x^3 + 1 - e^x = 0$.

```
> plot(x^3+1-exp(x), x=-3..5, -5..15);
```

El gràfic mostra **quatre** interseccions amb l'eix de les x . Una d'elles correspon a la solució que hem obtingut en l'Exemple 4.11. Sabrieu dir quina? Com trobaríeu les altres que falten?

Podem estendre la comanda `fsolve()` per a que miri de trobar una solució en un interval particular. Per exemple per a trobar la solució negativa li demanem a Maple que busqui en l'interval $[-1, -0.2]$ ja que podem veure a partir del gràfic que hi ha exactament una solució en aquest interval. La manera de fer-ho és escriure l'interval corresponent juntament amb el nom de la variable a la comanda `fsolve`.

```
> fsolve(eqn, x=-1..-0.2);
```

Per a determinar les altres dues solucions utilitzem `fsolve()` un altre cop, però aquesta vegada buscant a l'interval $[1, 2]$ i a l'interval $[4, 5]$.

```
> fsolve(eqn, x=1..2);
```

```
> fsolve(eqn, x=4..5);
```

Què passa si demaneu a Maple que busqui una solució en un interval on no hi ha solucions? Proveu-ho. A partir del gràfic és clar que no hi ha talls amb l'eix de les x entre 2 i 4 (i per tant no hi ha solucions).

```
> fsolve(eqn, x=2..4);
```

Noteu que Maple simplement respon amb la línia que originalment heu introduït sense cap canvi quan no pot trobar una solució a l'interval donat.

Hi ha altres solucions? Per exemple, hi ha alguna solució per a x més gran que 5? Podem comprovar això fent més gran l'interval sobre el que fem el gràfic. En la línia següent allarguem l'interval fins al $[-3, 50]$.

```
> plot(x^3+1-exp(x), x=-3..50, -10..15);
```

No apareixen nous talls amb l'eix de les x . El gràfic confirma el que podem esperar mirant els termes de l'equació, és a dir el terme exponencial domina i fa que el gràfic vagi baixant a la llarga.

Alternativament podem utilitzar la comanda `fsolve()`, ara buscant en aquest interval més gran.

```
> fsolve(eqn, x=5..50);
```

Tal com esperàvem Maple no haurà trobat solucions.

D'una forma semblant podem comprovar si hi ha solucions **cap a l'esquerra**. Aquí busquem si hi ha solucions a l'interval $[-50, -1]$.

```
> fsolve(eqn, x=-50..-1);
```

Cap ni una tampoc!

Finalment, hem pogut fer una llista completa de les solucions aproximades de la nostra equació original $x^3 + 1 - e^x = 0$. Són: -0.8251554597 , 0 , 1.545007279 i 4.567036837 .

Exemple 4.13

Utilitzem `fsolve()` per a calcular les solucions aproximades de l'equació: $\frac{x^2}{20} - 10x = 15 \cos(x + 15)$.

Com ja hem vist en l'últim exemple utilitzarem un gràfic per ajudar-nos a determinar el nombre i la situació aproximada de les solucions. La nostra feina es simplifica si comencem convertint l'equació que tenim en una d'equivalent que té com a part dreta un 0. Així buscarem les solucions de l'equació equivalent: $\frac{x^2}{20} - 10x - 15 \cos(x + 15) = 0$.

Si ara dibuixem el gràfic de la part esquerra d'aquesta equació obtindrem un altre cop solucions en cada un dels talls amb l'eix de les x .

```
> eqn:=x^2/20-10*x-15*cos(x+15)=0;
> plot(lhs(eqn),x=-10..10);
```

A partir del gràfic sembla que hi ha una solució a l'interval $[1, 2]$. Així doncs ara dirigirem Maple cap a trobar una solució en aquest interval.

```
> fsolve(eqn,x=1..2);
```

Hem trobat totes les solucions d'aquesta equació? De fet hi ha una altra solució! Per a trobar-la comenceu estirant l'interval sobre el que heu fet el gràfic. Aleshores utilitzeu `fsolve()` per a obtenir una aproximació numèrica d'aquesta segona solució.

Exercici 4.2

Determineu totes les solucions de l'equació $x^5 - 4x^3 + 3x^2 + 7x - 1 = 0$. Comenceu mirant un gràfic significatiu.

Exercici 4.3

Determineu totes les solucions de l'equació $x^2 - 2 = \ln(x + 5)$. Utilitzeu el gràfic **d'una** expressió per a localitzar les solucions. Comproveu cada una de les solucions substituint-la en l'equació original.

Exercici 4.4

Els gràfics de $y = 10 - x^2$ i de $y = 4 \sin(2x) + 5$ s'intersequen dues vegades sobre l'interval $[-5, 5]$.

- Feu el gràfic de les dues equacions juntes i estimeu amb el ratolí els punts d'intersecció.
- Escriviu una equació que es pugui resoldre per a determinar les coordenades x dels punts d'intersecció.

- c) Utilitzeu `fsolve()` per a resoldre aquesta equació.
- d) Utilitzeu els resultats de la part c) per a estimar les coordenades y dels punts d'intersecció.
- e) Sembla que les corbes es poden intersectar en un tercer punt a prop de $(1, 9)$. Utilitzeu `fsolve()` i/o un gràfic significatiu per a demostrar que no hi ha cap punt d'intersecció en aquest lloc.

4.4 Resoldre equacions formals

En moltes ocasions Maple també pot resoldre equacions de manera formal per a qualsevol de les variables que hi aparegui. Suposeu que volem obtenir la solució per a la variable g de l'equació: $4 - v = 2T - kg$. És a dir, volem aïllar g en funció de les demés variables. La comanda `solve()` funciona bé en aquest cas.

```
> solve(4-v=2*T-k*g,g);
```

Aquí hi ha una manera més maca per a mostrar el mateix resultat per pantalla:

```
> g=solve(4-v=2*T-k*g,g);
```

Exercici 4.5

Editeu l'última comanda per a trobar solucions per a les altres lletres T , k i v .

Exercici 4.6

Resoleu l'equació $x^2 + y^2 = 9$ per a la variable y . Assigneu el conjunt de solucions a una variable que es digui S . Quina relació hi ha entre les solucions $S[1]$ i $S[2]$?

4.5 Resoldre sistemes d'equacions lineals utilitzant la comanda `solve()`

Recordeu que cal reinicialitzar les variables abans de continuar. Carreguem també el paquet `plots`:

```
> restart:
```

```
> with(plots):
```

La comanda `solve()` també es pot utilitzar per a resoldre un sistema de m equacions lineals amb n incògnites. En direm un **sistema lineal m per n** perquè sigui més curt.

Exemple 4.14

Començarem per resoldre el sistema 2 per 2:
$$\begin{cases} 3x + 2y = 3 \\ x - y = -4 \end{cases}$$

```
> solve({3*x+2*y=3,x-y=-4});
```

Un gràfic de les dues funcions subjacents mostrarà que la solució correspon al punt d'intersecció en $(-1, 3)$. Però primer necessitem obtenir la forma explícita de cada una de les dues funcions lineals abans de poder fer-ne el dibuix. Per tant resolem cada una de les equacions respecte y .

```
> y1:=solve(3*x+2*y=3,y);
```

```
> y2:=solve(x-y=-4,y);
```

Ara construïm un gràfic format de dues parts: la “part1” conté els gràfics de les dues equacions i la “part2” dibuixa el punt que és la solució que hem trobat. Aquest punt ha de ser el punt d'intersecció de les dues línies. Ho és efectivament?

```
> part1:=plot([y1,y2],x=-5..5);
```

```
> part2:=plot([-1,3],style=point,color=blue,symbol=circle);
```

```
> display([part1,part2]);
```

Exemple 4.15

En aquest exemple solucionarem un sistema **3 per 3** amb incògnites x , y i z .

Resolem el sistema 3 per 3:
$$\begin{cases} x + y + z = 1 \\ 3x + y = 3 \\ x - 2y - z = 0 \end{cases}$$

```
> solve({x+y+z=1, 3*x+y=3, x-2*y-z=0});
```

Exercici 4.7

Determineu la solució del sistema:
$$\begin{cases} 4x + 3y = 12 \\ 5x - 7y = 35 \end{cases}$$

Comproveu la solució substituint els valors que s'obtenen en les dues equacions del sistema.

Sistemes lineals amb un nombre de solucions infinit

Quan un sistema té més **incògnites** que **equacions** sovint trobem no una sinó un nombre infinit de solucions. Tot seguit en veurem un exemple.

Exemple 4.16

Resolem el sistema:
$$\begin{cases} x + y + z = 1 \\ 3x + y = 3 \end{cases}$$

```
> solns:=solve({x+y+z=1, 3*x+y=3});
```

Noteu que aquest cop no obtenim un únic conjunt de valors numèrics per a x , y i z . En lloc d'això Maple ens diu com han d'estar relacionats els diferents valors de x , y i z per a construir una solució típica.

En particular l'expressió $y = y$ en la resposta anterior indica que la incògnita y pot ser **qualsevol** número. Ens referirem a aquesta incògnita com la *variable lliure* de la solució. Per a obtenir qualsevol **solució particular** (entre les infinites solucions que hi ha) trieu un valor per a la y i utilitzeu-lo per a calcular els valors corresponents de la x i de la z . Per exemple si $y = -9$.

```
> subs(y=-9,solns);
```

Així una solució particular és: $x = 4$, $y = -9$ i $z = 6$.

Preneu-vos un minut de temps i verifiqueu a mà que aquests tres nombres satisfan realment les equacions originals: $x + y + z = 1$ i $3x + y = 3$.

Ara mirem la solució generada quan prenem $y = -3$.

```
> subs(y=-3,solns);
```

Així doncs dues de les infinites solucions que obtenim per a les incògnites (x, y, z) són $(4, -9, 6)$ i $(2, -3, 2)$.

Exercici 4.8

Resoleu el sistema: $\{x + 2y + z = 2, 3x + y = 1\}$ i doneu com a mínim tres solucions particulars.

5 Funcions: definir, avaluar i fer-ne gràfics

L'objectiu bàsic d'aquesta secció és el d'aprendre primer de tot a definir funcions $f(x)$. Després també veurem com avaluar funcions, resoldre equacions on hi intervenen funcions, i com fer-ne gràfiques.

```
> restart;
```

5.1 Definir i esborrar una funció en Maple

Per a distingir entre una funció i una expressió, Maple utilitza una notació especial quan es defineix una funció. Per exemple, considerem la funció $f(x) = \cos(\pi x) + 3$. En Maple s'introdueix com:

```
> f:=x->cos(Pi*x)+3;
```

És important que preneu nota de la sintaxi que utilitzem. És **absolutament necessari** introduir la fletxa “->” construïda amb el signe “menys” i el símbol “més gran que”. **Maple no definirà una funció** si introduïu $f(x) := \cos(\pi x) + 3$.

A continuació podeu veure la diferència entre una expressió i una funció en Maple. Fixeu-vos amb la diferència de la sintaxi i en com dóna el resultat Maple per a cada un dels dos casos.

```
> y:=(x + 2)/(x^3 + 5*x + 2);
```

```
> f:=x->(x + 2)/(x^3 + 5*x + 2);
```

Les funcions necessiten sempre una fletxa quan s'han d'introduir; en el resultat que dóna Maple també hi ha d'haver una fletxa. Verifiqueu sempre que en el resultat hi ha la fletxa per a confirmar que heu definit una funció.

Exercici 5.1

Definiu la funció $h(x) = x^3 \sin(2x + 1)$.

Evidentment, no cal que les funcions depenguin d'un sol argument. Podem considerar per exemple

Exemple 5.1

```
> f := (x,y)->x^2-x*y+ln(x^2+y^2);
```

Aquí obtenim una funció f que admet dues variables (x,y) .

Quan ja heu definit una funció, Maple recorda aquesta funció durant tota la sessió de treball. Si voleu substituir la funció per una nova definició, simplement reescriuiu la definició. Per exemple, si voleu substituir la funció anterior $f(x,y)$ per $\ln(\cos(5x))$, escriviu:

```
> f:=x->ln(cos(5*x));
```

Podeu confirmar el valor actual de la funció $f(x)$:

```
> f(x);
```

És important que noteu que mentre f es una funció, el valor de $f(x)$ és una expressió i es pot tractar com a tal. És a dir, podem aplicar-li qualsevol de les comandes que hem vist per expressions. Per exemple, avaluant,

```
> subs(x=0,f(x));
> subs(x=2*t+1,f(x));
```

Ara bé, aquesta no és la millor manera d'avaluar una funció com veurem més endavant.

Si voleu esborrar la funció f sense definir-la de nou utilitzem el mateix procediment que el que vam veure a la secció 1 per esborrar variables, és a dir, escriviu:

```
> f:='f';
```

Sempre és una bona idea esborrar les funcions que tingueu quan comenceu un problema nou. De forma alternativa també podeu utilitzar la comanda `restart` per a netejar tot el que hi hagi en la memòria de la sessió.

5.2 Avaluar una funció

Quan heu definit una funció, podeu avaluar-la per a diferents valors o per a expressions literals utilitzant la notació funcional. És a dir, com veurem en l'exemple següent podem substituir x per l'expressió que volgüem. Abans de definir la funció però netegem el valor de f .

Exemple 5.2

```
> f:='f';
> f:=x->3*x+x^2;
> f(-1);
> f(2+sqrt(5));
> evalf(f(2+sqrt(5)));
> f(x+4);
> simplify(%);
> (f(x+h)-f(x))/h;
> simplify(%);
```

Si intervenen més d'una funció, la composició de funcions és fàcil de fer. Simplement, consisteix en aplicar la definició de composició de funcions.

```
> g:=x->cos(x)+1;
> f(g(Pi/3));
```

```

> g(f(Pi/3));
> j:=x->g(f(x));
> j(x);
> j(Pi/3);

```

Exercici 5.2

Definiu les funcions $s(t) = \frac{3+t^2}{\sqrt{3t+1}}$ i $k(x) = (x+1)^2$ i feu que Maple calculi $s(2)$, $s(t-3)$, $s(t) - s(3)$ i $s \circ k(3)$ simplificant els resultats. No oblideu la notació de la fletxa!

5.3 Resoldre equacions en les que intervenen funcions

Un cop definida una funció, també podem resoldre equacions amb aquesta funció de manera exacta o aproximada utilitzant les comandes vistes a la secció 4.

```

> g:='g';
> g:=t->t^3-6*t^2+6*t+8;
> solve(g(t)=0,t);
> fsolve(g(t)=0,t);

```

Com que g ha estat definida com una funció i no una expressió, el nom de la variable no és important. Això no passava quan utilitzàvem expressions en la secció anterior.

```

> fsolve(g(x)=0,x);

```

5.4 Gràfics de funcions

La comanda `plot` funciona igualment per a funcions f . Sempre podem utilitzar la comanda `plot` amb l'expressió obtinguda avaluant a la variable x , $f(x)$. Vegem-ho en el següent exemple.

Exemple 5.3

```

> h:='h'; x:='x';
> h:=x->x*exp(-x);
> plot(h(x),x=-1..4,-2..1);

```

Es poden dibuixar gràfics de diferents funcions simultàniament de la mateixa manera que es fa per a les expressions.

Considerem la funció $f(x) = \frac{2}{x^2 + 1}$. A continuació fem el gràfic d'aquesta funció i dels seus desplaçaments horitzontals $f(x + 1)$, $f(x - 3)$ i $f(x - 6)$. Podeu identificar cada un d'aquests gràfics?

```
> f:=x->2/(x^2+1);
> f(x+1);
> f(x-3);
> f(x-6);
> plot([f(x),f(x+1),f(x-3),f(x-6)],x=-5..10,-1..3);
```

La comanda `plot` admet també que l'entrada sigui una funció directament f . Aleshores no s'ha d'especificar la variable x en el rang.

```
> plot(f,-5..10,-1..3);
```

Exercici 5.3

Definiu la funció $f(x) = 2x - |x^2 - 5|$ i tot seguit responeu a les qüestions següents:

- Quin és el valor de $f(6.5)$
- Simplifiqueu el valor de $f(z - 4)$, on z és una variable.
- Feu el gràfic de $f(x)$.
- Determineu els valors de x per als que $f(x) = 0$.

Exercici 5.4

Definiu les funcions $g(x) = 5e^{x/2}$ i $h(x) = x + 10$ i feu el següent:

- Dibuixeu un gràfic que mostri les dues funcions $g(x)$ i $h(x)$. Experimenteu amb diferents valors per al domini i per al recorregut.
- Feu una estimació del punt d'intersecció d'aquests dos gràfics utilitzant el botó esquerra del ratolí.
- Utilitzeu la comanda `fsolve()` per a resoldre l'equació $g(x) = h(x)$. Com es relaciona la solució d'aquesta equació amb el que heu obtingut en l'apartat (b)?

Exercici 5.5

Definiu la funció $k(x) = x + 3\sin(2x)$, després feu el següent:

- Dibuixeu el gràfic d'aquesta funció en el domini $[-1, 8]$.
- Modifiqueu el dibuix de l'apartat (a) per tal que inclogui la línia horitzontal $y = 4$. Utilitzeu aquest nou gràfic per a estimar el nombre i els valors aproximats dels x tals que $k(x) = 4$.
- Quina funció única hauríeu de dibuixar per a obtenir la mateixa informació que en l'apartat (b)?

d) Utilitzeu la comanda de Maple `fsolve()` per a aproximar totes les solucions de l'equació $k(x) = 4$.

5.5 Com convertir expressions en funcions

Fins ara hem vist com a partir d'una funció f definida en Maple, la podem convertir en una expressió mitjançant avaluació en una variable $f(x)$. Maple també conté la comanda que ens permet resoldre el problema invers, és a dir, donada una expressió ens pot ser molt útil convertir-la en una funció. Aquesta situació es resol amb la comanda `unapply()`. Observeu el seu funcionament en l'exemple següent.

Exemple 5.4

```
> f:='f';
> expr:=ln(3*x^3+sqrt(x-1));
> f:=unapply(expr,x);
> f(x);
> evalf(f(3));
```

Podeu veure que la comanda `unapply()` té dos arguments: el primer és l'expressió que volem convertir en funció i el segon és la variable que surt en l'expressió respecte de la que volem que depengui la funció resultant.

Exercici 5.6

Considerem l'expressió $x^2 - y^2 + \sin(xy)$. Mireu en l'ajuda sobre `unapply` la manera de convertir l'expressió anterior en una funció dependent de les dues variables x i y .

Exemple 5.5

Sigui $y = f(x)$ una funció tal que satisfà la relació $yx = -y^2 + 7x + 2y$. Volem definir la corresponent funció f dependent de x . Primer utilitzem la comanda `solve` per aïllar y en funció de x i després convertim l'expressió resultant (o expressions resultants) en funcions.

```
> expre:=y*x=-y^2+7*x+2*y;
> yexpre:=solve(expre,y);
> f1:=unapply(yexpre[1],x);
> f2:=unapply(yexpre[2],x);
> plot([f1,f2],-1..2);
```

Exemple 5.6

En algunes ocasions és millor conservar les expressions mitjançant funcions per evitar perdre la informació que conté l'expressió quan es manipulen les variables. Per exemple, considerem l'expressió següent i la seva corresponent funció,

```
> ex:=(ln(x+1)-x^2)/(x+1);  
> func:=unapply(ex,x);
```

Durant la sessió de Maple podem assignar un valor a la variable x . Aleshores observeu que passa amb l'expressió i la funció.

```
> x:=3;  
> ex;  
> func(x);
```

Ja no podem recuperar l'expressió inicial utilitzant ex (sense haver de *buidar* la variable x) que ha passat a tenir un valor numèric però si que podem recuperar-la mitjançant la funció que hem definit.

```
> func(t);
```

Exercici 5.7

Sigui $y = f(x)$ una funció que satisfà la igualtat $yx + y^2 - 6x + x^2 = 0$. Mitjançant la comanda `solve` doneu les expressions de y en funció de x i convertiu-les en funció. Feu-ne les corresponents representacions gràfiques.